리졸빙
 kbmMW v. 2.50+
Pro, ProPlus and Enterprise Editions
소개

쿼리 서비스를 통해서 쿼리의 결과셋을 받을 수 있지만, 결과셋에 변경된 내용을 뒷단의 데이터저장소 (데이터베이스)에 있는 원래 데이터가 있는 테이블(보통은 그렇습니다)로 적용하는 것도 중요합니다. 

이 개념을 kbmMW에서는 ‘리졸빙’이라고 합니다.

리졸빙이라는 말이 쉽게  들리겠지만,  실제로는 꽤 복잡하고 여러 변수들이 있습니다. kbmMW는  복잡한 부분을 숨기고 개발자들이 대단히 복잡한 상황에서도 쉽게 데이터를 리졸브할 수 있게 해줍니다.

이 문서에서는 쉬운 것부터 어려운 순서로 여러 가지 리졸빙 상황들을 설명합니다. 가장 잘 이해하려면, 이 문서 전체를 읽을 필요가 있습니다.

1티어 리졸빙

쉬운 경우부터 시작하기 위해, 데이터베이스 어댑터 컴포넌트들만을 사용하는 표준 TForm 기반 애플리케이션을 만들어보겠습니다. 이 애플리케이션의 목적은 BDE DBDEMOS의 테이블 customer에서 모든 레코드를 셀렉트하고 kbmMW BDE 어댑터 컴포넌트들을 이용하여 사용자가 그 내용을 수정할 수 있게 하는 것입니다.

이 예제는 하찮아 보이겠지만, kbmMW에서의 리졸빙에 대한 여러 가지 기본적인 측면들을 보여줍니다. 이후에 우리는 n티어 리졸빙을 하기 위해 같은 지식을 이용할 것입니다.

TkbmMWBDEConnectionPool, TkbmMWPooledSession, TkbmMWBDEQuery, TkbmMWBDEResolver 컴포넌트가 하나씩 필요합니다.  이들 컴포넌트의 사용법에 대해 더 자세히 알아보려면 “kbmMW 쿼리 서비스 개발” 문서를 참고하십시오.

추가로, 리졸버에는 데이터를 리졸브할 데이터베이스의 타입에 대한 정보도 필요합니다. 이런 목적으로 메타데이터 컴포넌트가 필요합니다. 이번 예제의 경우에는 TkbmMWGenericSQLMetaData를 선택하겠습니다. 자동증가 필드가 리졸브되고 그 값들이 사용자에게 정확히 돌아가기를 원하는 경우라면, 더 정확한 메타데이터 컴포넌트를 선택하는 것이 특별히 중요합니다.
리졸버 컴포넌트의 책임은 데이터셋의 변화를 데이터베이스/데이터저장소로 반영하는 방법을 관리하는 것입니다. 따라서 리졸버 타입에 따라 여러 속성들을 가지고 있습니다.

kbmMW는  SQL 및 비 SQL 데이터베이스/데이터저장소 양쪽 모두 지원하며, 따라서 그만큼 여러 가지 종류의 리졸버들이 있습니다. SQL 리졸버는 SQL  리졸빙 관련의 속성들을 가지고 있습니다.

다른 리졸버 타입들은 각각의 동작 방식을 제어하기 위한 다른 속성들을 가지고 있습니다. 우리는 많은 사람들이 가장 많이 사용하는 SQL 기반 리졸빙을 주로 살펴볼 것입니다.

리졸버는 쿼리 컴포넌트와 밀접하게 상호 연동하여 동작합니다. 쿼리 컴포넌트는 결과셋에 일어나는 모든 딜리트, 업데이트, 인서트 동작들을 추적, 관리할 책임이 있습니다.

쿼리 컴포넌트는 레코드 버저닝을 통해 이런 작업을 합니다. 사용자가 한 레코드를 딜리트하면 원래의 레코드는 실제로는 계속 존재하지만 삭제된 것으로 표시됩니다. 사용자가 레코드를 수정하면 원래의 레코드는 그대로 유지되지만 새로운 수정된 버전이 추가됩니다.

따라서 쿼리 컴포넌트의 데이터 변경을 리졸브할 수 있으려면 쿼리 컴포넌트가 EnableVersioning:=true로 설정되어야 합니다. (기본값입니다)

VersioningMode는 두 가지 값을 가질 수 있지만, mtvm1SinceCheckPoint가 메모리를 적게 소모합니다. 다른 모드(mtvmAllSinceCheckPoint)는 애플리케이션에서 여러 단계의 실행취소(undo)를 지원하려고 할 때 유용합니다. (for example using StartTransaction, Commit and Rollback, or using the Undo method)
쿼리 컴포넌트를 리졸빙을 위해 설정하려면, Resolver 속성이 반드시 같은 타입의 리졸버로 설정되어야 합니다.
애플리케이션

이제 다음 그림과 같이 간단한 예제 애플리케이션을 만들어봅시다.

다음과 같이 속성들을 설정합니다.

kbmMWConnectionPool1.Database:=Database1
kbmMWConnectionPool1.MetaData:=kbmMWGenericSQLMetaData1
Database1.AliasNames:='DBDEMOS'
kbmMWPooledSession1.ConnectionPool:=kbmMWConnectionPool1
kbmMWPooledSession1.SessionName:='DEMO'
kbmMWBDEQuery1.SessionName:='DEMO'
kbmMWBDEQuery1.SQL.Text:='select * from customer'
kbmMWBDEQuery1.Resolver:=kbmMWBDEResolver1
DataSource1.DataSet:=kbmMWBDEQuery1
DBGrid1.DataSource:=DataSource1

Query 버튼의 이벤트 핸들러에는 다음과 같이 코딩합니다.

kbmMWBDEQuery1.Open;

Resolve 버튼의 이벤트 핸들러에는 다음과 같이 코딩합니다.

kbmMWBDEQuery1.Resolve;
이 예제에서 쿼리는 BDE 기반의 쿼리이기 때문에 TkbmMWBDEQuery 컴포넌트를 사용하며, 따라서 데이터를 리졸브하기 위해서는 TkbmMWBDEResolver 컴포넌트를 선택했습니다.

리졸버가 뒷단의 데이터베이스를 위한 정확한 SQL 문을 생성하기 위해서는 정확한 메타데이터 컴포넌트를 선택해서 설정해야 합니다. 몇가지 메타데이터 컴포넌트가 있지만, 일부는 특정한 데이터베이스를 위해 설계된 것이며, 지금 우리가 선택한 TkbmMWGenericSQLMetaData와 같은 나머지는 더 일반적인 것으로서, 데이터베이스와 맞추기 위해 여러 설정들을 수정할 수 있게 되어 있습니다.

DateLayout – 데이터베이스에 맞는 날짜 포맷을 지정합니다. 일반적으로 파라미터 기반의 리졸빙에서는 사용되지 않습니다.(대부분의 리졸버가 그렇습니다)

DateTimeLayout – 데이터베이스에 맞는 날짜/시간 포맷을 지정합니다. 일반적으로 파라미터 기반의 리졸빙에서는 사용되지 않습니다.(대부분의 리졸버가 그렇습니다)

TimeLayout - 데이터베이스에 맞는 시간 포맷을 지정합니다. 일반적으로 파라미터 기반의 리졸빙에서는 사용되지 않습니다.(대부분의 리졸버가 그렇습니다)

FalseValue – 데이터베이스에 맞는 불린 false 값의 포맷을 지정합니다. 일반적으로 파라미터 기반의 리졸빙에서는 사용되지 않습니다.(대부분의 리졸버가 그렇습니다)

TrueValue – 데이터베이스에 맞는 불린 true 값의 포맷을 지정합니다. 일반적으로 파라미터 기반의 리졸빙에서는 사용되지 않습니다.(대부분의 리졸버가 그렇습니다)

FieldNameBrackets – 필드네임이 괄호 등으로 싸여야 하는지 여부를 지정합니다. 이 값이 true이면 FieldNameQuote 속성에서 시작 괄호 종류를 지정합니다. ({ [ < ( 중 하나). 끝나는 괄호는 KbmMW에서 자동으로 선택됩니다.
FieldNameCase – Unaltered는 필드네임을 그대로 사용한다는 의미입니다. Lower는 필드네임을 소문자로 변환한다는 의미입니다. Upper는 필드네임을 대문자로 변환한다는 의미입니다. SQL 생성에 직접적으로 영향을 미칩니다.
FieldNameQuote – 필드네임을 감쌀 문자를 지정합니다. FieldNameBrackets가 true이면, FieldNameQuote는 (, {, [, < 중의 하나여야 하며 이것은 괄호의 시작 문자를 의미합니다. 끝나는 괄호 문자는 KbmMW가 자동으로 선택합니다.
PrependTableName –필드네임 앞에 테이블 이름이 붙여져야 하는지 여부입니다.
QuoteAllFieldNames – true이면 모든 필드네임을 FieldNameQuote 및 FieldNameBrackets 설정에 따라 형식을 만듭니다. False이면 부적절한 문자를 포함한 필드네임만 괄호로 싸여집니다.

QuoteTableName – true이면 테이블 이름을 TableNameQuote 문자로 쌉니다.
QuoteStringQuote – 이 속성은 StringQuote 문자가 실제 문자열에 포함되어 있을 경우 그 앞에 추가될 문자를 지정합니다. 예를 들어, QuoteStringQuote = \ 이고 StringQuote=’ 인 경우, The lady’s purse 와 같은 텍스트는 ‘The lady\’s purse’ 처럼 되게 됩니다.

TableFieldSeperator – 테이블 이름과 필드 네임이 조합될 때 사용될 문자를 지정합니다. 일반적으로 마침표(.)가 사용됩니다.
TableNameBrackets – 테이블에 적용된다는 점을 제외하면 FieldNameBrackets와 같습니다.
TableNameCase – 역시 FieldNameCase와 비슷합니다.

TableNameQuote - 테이블네임을 감쌀 문자를 지정합니다. TableNameBrackets가 true이면, TableNameQuote는 (, {, [, < 중의 하나여야 하며 이것은 괄호의 시작 문자를 의미합니다. 끝나는 괄호 문자는 KbmMW가 자동으로 선택합니다. TableNameQuote가 #0 (chr(0))으로 설정되면 테이블네임을 전혀 감싸지 않습니다.

StringQuote – 문자열을 감쌀 때 사용될 문자를 지정합니다. 문자열 내에 이 문자가 나타날 때 어떻게 처리할 지에 대해서는 QuoteStringQuote를 참고하십시오.

메타데이터 컴포넌트에 대해 더 자세히 알고 싶으면, ‘kbmMW and generalized metadata management’ 문서를 참고하십시오.

리졸버에는 리졸빙 과정을 제어하기 위한 몇가지 설정들이 있습니다.

AllOrNothing – 모든 레코드들이 단일 작업으로 리졸브되어야 할 경우 true로 설정합니다. 레코드 작업들 중 하나가 어떤 이유로 뒷단의 데이터베이스나 데이터 저장소로 리졸브될 수 없는 경우, 모든 변경 내용들이 롤백됩니다. false로 설정하면 에러가 발생한 레코드와는 관계없이 나머지를 리졸브합니다. 개발자가 이 프로퍼티를 true로 설정할 경우가 종종 있습니다.

IgnoreFailedDelete – true이면 레코드가 삭제될 수 없을 때 리포트하지 않습니다.
IgnoreFailedInsert – true이면 레코드가 인서트될 수 없을 때 리포트하지 않습니다.
IgnoreFailedUpdate - true이면 레코드가 수정될 수 없을 때 리포트하지 않습니다.
InsertKeyFields – true이면 키 필드의 값들도 인서트합니다.

InsertOnFailedUpdate – true이면 레코드가 수정될 수 없을 때 새 레코드로 인서트하려고 시도합니다.
SkipFieldsWithoutOrigin – true이면 빈 Origin 값을 가진 필드는 리졸브 작업에 포함되지 않습니다.
UpdateInsertAutoIncFields – true이면 ftAutoInc 타입의 필드가 업데이트와 인서트 문에 포함됩니다.
UpdateInsertReadonlyFields – true이면 ReadOnly인 필드가 업데이트와 인서트 문에 포함됩니다.
UpdateKeyFields – true이면 레코드의 유니크 키에 포함된 필드도 업데이트 문에 포함됩니다.

UpdateWhereAll – true이면 SQL의 where 절에 모든 필드를 포함시킵니다. False이면 정의된 키 필드들만 포함시킵니다.

이 애플리케이션을 실행시키고 Query 버튼을 클릭하면 DBGrid에 많은 레코드들을 볼 수 있을 것입니다. 내부적으로는 커넥션풀에 의해 DBDEMOS 데이터베이스에 대한 커서/연결이 오픈되고, 이렇게 오픈된 연결이 쿼리 컴포넌트에게 예약되고, 쿼리가 실행되고, 모든 레코드가 kbmMWBDEQuery 컴포넌트로 페치되는 것입니다. 여기서 사용된 연결은 닫히는 대신 이후에 다시 재사용되기 위해 커넥션풀로 되돌려집니다.

DBGrid에서 데이터를 수정하더라도 데이터베이스의 데이터가 바로 수정되지는 않는데, 그것은 쿼리 컴포넌트가 더 이상 데이터베이스에 연결되어 있지 않기 때문입니다. 이런 개념을 끊어진(disconnected) 데이터셋이라고 합니다.

우리는 삭제, 인서트 등 우리가 원하는 어떤 수정 작업이든 레코드에 할 수 있지만 이 모든 변경은 DBDEMOS 데이터베이스에게는 알려지지 않습니다. 우리가 그런 변경들을 리졸브하기 전까지는 말입니다.

변경 내용을 리졸브한다는 것은, 리졸버 컴포넌트가 변경, 인서트, 딜리트된 내용이 있는지 쿼리 컴포넌트의 recordstorage를 확인하는 것을 의미합니다. 리졸버는 이런 변경 내용들을 ‘INSERT …VALUES….’, ‘DELETE…. WHERE…’, ‘UPDATE…. SET....’과 같이 파라미터를 가진 SQL문을 만들고 실행하여 데이터베이스에 반영하려고 시도합니다.

정확한 SQL 문을 생성하기 위해, 리졸버는 데이터를 받을 테이블의 이름을 알아야 합니다. 우리는 그것이 ‘customer’ 테이블이라는 것을 알고 있습니다. 리졸버는 이 정보를 쿼리 컴포넌트의 TableName 프로퍼티로부터 알아냅니다. 따라서 kbmMWBDEQuery.TableName:=’customer’ 으로 설정하십시오.

리졸버 컴포넌트가 알아야 할 다른 한가지는 유니크 키를 구성하는 필드들의 이름입니다. 우리는 유니크 키가 ‘custno’라는 것을 알고 있습니다. 따라서 kbmMWBDEQuery.KeyFieldNames:=’custno’ 으로 설정하십시오.

유니크 키가 여러 필드로 구성되어 있으면, 필드 이름들을 세미콜론(;)으로 나누어야 합니다. 예를 들면 ‘field1;field2’ 이렇게 말입니다.

이제 데이터베이스로 변경 내용을 리졸브하기 위한 모든 준비가 끝났습니다.

적절한 시점이 되었거나 Resolve 버튼이 클릭되었을 때 쿼리 컴포넌트의 Resolve 메소드를 호출하기만 하면 됩니다. 

에러 처리

그렇다면, 어떤 이유로든 리졸브 작업의 일부 혹은 전체가 완수될 수 없을 때는 어떻게 될까요?

리졸브 과정 동안, 리졸버는 에러 테이블(kbmMWBDEQuery1.ErrorTable 속성을 통해 액세스할 수 있습니다)을 구축합니다. 에러 테이블은 리졸브될 수 없는 모든 레코드들에 대한 레퍼런스를 가지고 있으며, Ignorexxxx 속성 등의 리졸버의 설정이 어떻든 관계 없이 모든 에러가 등록됩니다.

에러 테이블은 4개의 표준 필드를 가지고 있으며 아래와 같이 상수로 정의되어 있습니다.

  KBMMW_ERROR_RESOLVER_RECORDID_FIELDNAME = 'KBMMW_RECORDID';
  KBMMW_ERROR_RESOLVER_TYPE_FIELDNAME = 'KBMMW_ERRORTYPE';
  KBMMW_ERROR_RESOLVER_MESSAGE_FIELDNAME = 'KBMMW_ERRORMESSAGE';
  KBMMW_ERROR_RESOLVER_DATA_FIELDNAME = 'KBMMW_DATA';

리졸브 직후에 바로 ErrorTable 을 액세스할 수 있으며 리졸브 과정에서 에러가 발생했는지 알아보기 위해 레코드들을 확인할 수 있습니다. ‘KBMMW_ERRORMESSAGE’ 필드에서 원인을 설명하는 메시지를 볼 수 있을 것입니다.

만약 ‘KBMMW_ERRORTYPE’ 필드의 내용이 0 값(kbmMWErrorResolverDB)을 가지고 있다면 해당 레코드에 대해 리졸브가 성공하지 못한 것입니다. 이 필드에는 다른 값들도 있는데, 1값(kbmMWErrorResolverModified)은 실제로는 에러가 아니며 리졸브하는 동안 필드의 값이 변경되었다는 것을 의미하며, 사용자에게 그런 사실을 알리는 것입니다. 리졸브가 실패했다는 의미는 아닙니다.

쿼리의 레코드들 중에서 리졸브되지 못한 레코드를 찾기 위해서는 SearchRecordID 메소드를 이용하여 RecordID를 찾아가면 됩니다. 이 메소드는 요청한 레코드의 recordnumber를 리턴합니다. 예를 들면 다음과 같습니다.

var

  i: integer;

begin

  q.CurIndex.SearchRecordID(q.ErrorTable.FieldByName('KBMMW_RECORDID').AsInteger, i);

  if (i<0) then Raise Exception.Create('Record not found');

  q.RecNo := i+1;

  // Now the failing record has been found.

  ShowMessage('Record with value '+q.Fields[0].AsString+' didn''t resolve.');

end;

KBMMW_RECORDID 필드가 레코드를 구분합니다. 클라이언트 쿼리에서 리졸브를 요청한 경우에는 레코드ID는 클라이언트 레코드의 레코드ID이며, 아닌 경우에는 서버쪽 쿼리 레코드의 레코드ID입니다.

레코드ID 없이 에러 레코드가 등록되었을 수도 있습니다. 이것은 에러가 특정 레코드에 관련된 에러가 아닌 경우입니다.

조금 덜 번거롭고 쉬운 방법은 쿼리나 스토어드 프로시저 컴포넌트의 OnResolveError 이벤트의 핸들러를 만들어 코딩하는 것입니다. 이 이벤트는 리졸브가 실패한 각각의 레코드마다 못할 때마다 호출되며, 에러 타입(항상 kbmMWErrorResolverDB)과 에러 메시지에 대한 정보들을 인자로 받습니다.
가능하다면, 

If possible, it automatically makes sure that the current record in the query/storedprocedure dataset, is the failing record and thus information about the record can be obtained through normal field operations. Check the Current argument to see if a current record could be determined.

Finally the OnResolveError event have a Retry argument which is default true. It can be set to false if this particular record should not be re-resolved on next resolve operation. All failing records are normally automatically marked to be resolved again on next opportunity.

Original field values for the current record can be obtained like this:

var
  oldvalue: variant;
  n: integer;
  us: TUpdateStatus;
begin
  // Get the original value for a field as a variant.
  oldvalue := q.GetVersionFieldData(q.FieldByName('somefieldname'), 9999);

  // oldvalue contain the value of the field 9999 versions ago…
  // which in reality means the original value.
  // The number of versions of the record can be obtained by
  n := q.GetVersionCount;

  // And the updatestatus of a particular version can be obtained by:
  us := q.GetVersionStatus(x);

  // where x is the version number. 0=current version, 1=older version,
  // 2=even older version etc.
end;

Raising errors manually

On occasions, it’s practical to validate the resolve operation while it’s taking place, and potentially stop it or signal a problem back to the user.

Let’s say you want to check the contents of a new record before its being inserted into the database.

Write some code for the OnInsert event handler of the resolver component. E.g

procedure TForm1.kbmMWBDEResolver1Insert(Sender: TkbmMWCustomResolver;

  var Skip, Abort: Boolean);

begin

  if Sender.ValueByFieldName['afieldname']=22 then
    Skip := true;

end;

What this does is that it checks the value of the field named ‘afieldname’ for its value. If it’s 22, we ask the resolver to skip the insert silently.

We could also have forced an abort of the resolving operation by setting Abort:=true.

The client would immediately get to know that the resolve operation was aborted. However the user wouldn’t get any information that may have been provided in the errortable by previous records being resolved in the same operation.

If we would like to signal a problem with a specific record in a way that the user would be able to receive via the error table, and without aborting the resolve operation as such, we can use the LogError and LogRecordError methods. E.g.

procedure TForm1.kbmMWBDEResolver1Insert(Sender: TkbmMWCustomResolver;

  var Skip, Abort: Boolean);

begin

  if Sender.ValueByFieldName['afieldname']=22 then
  begin

    Sender.LogRecordError('We don'''t like the value 22 in afieldname');

    Skip := true;

  end;

end;

LogRecordError logs an error in the errortable along with a record reference, in such way that the record in question can be selected automatically for the user to scrutinize.

If there are errors that can’t really be related to a specific record, instead use LogError.
리졸브 작업중에 값을 변경하기
개발자는 리졸빙 과정이 진행되기 직전이나 리졸빙 작업중에 리졸버가 작업하는 값들을 변경할 수도 있습니다.

리졸버는 OnResolveInitialization 라는 이벤트를 가지고 있습니다. 이 이벤트는 dataset.Resolve 메소드를 호출하는 등 리졸버 작업이 시작되려고 할 때마다 발생합니다. 

이 이벤트 동작중에 리졸브되려는 전체 데이터셋의 값들을 변경하는 것이 가능합니다. 리졸버의 Dataset 프로퍼티를 통해 해당 데이터셋을 액세스할 수 있습니다.

기억할 것은, 여러분이 이 시점에서 수정하는 모든 것들은 리졸빙 작업에 직접적으로 반영된다는 것입니다. 예를 들어, 데이터셋에서 레코드 하나를 삭제한다면, 리졸버는 뒷단의 데이터베이스로부터 해당 레코드를 삭제하려고 시도합니다.

따라서 OnResolveInitialization 이벤트는 전체 데이터셋에 대해 작업할 수 있도록 해줍니다. 하지만 필드를 추가하거나 삭제하려고 시도하지는 말아야 합니다. 데이터셋 자체는 변경되어서는 안되기 때문입니다.

리졸빙 작업중에 데이터셋으로부터 리졸버가 가져오는 데이터를 변경하는 다른 방법은, 리졸버의 OnGetValue 이벤트를 통하는 것입니다.

이 이벤트는 현재 레코드의 리졸빙 작업에 대해 특정 필드에 새 값을 할당할 수 있게 해줍니다. 예를 들어,

procedure TForm1.kbmMWBDEResolver1GetValue(

  ADeltaHandler: TkbmCustomDeltaHandler; AField: TField;

  var AValue: Variant);

begin

  if AField.FieldName='field1' then
    AValue:=22;

end;

위의 경우, 리졸버는 field1이라는 이름의 필드에 대해 언제나 22라는 값을 얻게 됩니다. 이런 코드는, 여러분이 어떤 값이 필요한지를 알고 있을 때가 아니면 상대적으로 덜 사용될 것입니다. 특정 고정된 값을 할당하는 것은 보통 인서트 동작이므로, 업데이트 동작에서는 데이터셋 자체의 원래 값을 그대로 유지하려고 할 수 있습니다.

procedure TForm1.kbmMWBDEResolver1GetValue(

  ADeltaHandler: TkbmCustomDeltaHandler; AField: TField;

  var AValue: Variant);

begin

  if (TkbmMWCustomResolver(ADeltaHandler).State=mwrsInsert) and
     (AField.FieldName='field1') then
    AValue:=22;

end;

위의 수정된 코드에서는 레코드 작업이 인서트인지를 검사합니다. 리졸브 작업중에 그런 경우에만 field1 필드에 22 값을 할당하고 있습니다.

필요하면 언제든지 리졸버의 상태가 mwrsInsert, mwrsDelete, mwrsModify, mwrsIdle(리졸버가 리졸빙 작업을 하고 있지 않을 때) 중 어느 것인지 검사할 수 있습니다.
좋습니다. 이제 우리는 리졸빙 작업중에 새 값을 지정할 수 있게 되었습니다. 하지만 이것이 우리 사용자들을 혼란스럽게 하지는 않을까요? 사용자는 자신의 연결이 끊어진(disconnected) 데이터셋에서 어떤 값을 보고 있는데, 데이터베이스는 다른 값을 가지게 됩니다.

따라서 변경이 일어난 값에 대해 사용자에게 알려줄 필요가 있습니다. 이것은 LogFieldValueChange 메소드로 쉽게 구현할 수 있습니다. 다음 예를 보시기 바랍니다.

procedure TForm1.kbmMWBDEResolver1GetValue(

  ADeltaHandler: TkbmCustomDeltaHandler; AField: TField;

  var AValue: Variant);

begin

  with TkbmMWCustomResolver(ADeltaHandler) do
  begin

    if (State=mwrsInsert) and (AField.FieldName='field1') then
    begin

      Avalue := 22;

      LogFieldValueChange([AField],[AValue], 'We chose 22 instead for field1');

    end;

  end;

end;

LogFieldValueChange 메소드는 세개의 인자를 갖는데, 데이터셋에서 변경되어 리졸브된 필드들의 배열, 그 각각에 매치되는 새 값들을 의미하는 variant 배열, 그리고 마지막으로 여러분이 표시하고 싶은 메시지입니다.

기본적으로는, 이 메소드를 통해 기록되는 변경들은 자동으로 원래의 데이터셋에 업데이트됩니다. 따라서 사용자는 리졸브 작업이 끝나자마자 새 값들을 보게 되는 것입니다.

사용자에게 변경의 내용을 알리고 싶다면 데이터셋의 OnResolveFieldValueChange 이벤트를 이용하면 됩니다. 예를 들어,
procedure TForm1.kbmMWBDEQuery1ResolveFieldValueChange(Sender: TObject;

  Fields: array of TField; Message: String; RecordID: Integer;

  Current: Boolean; var Skip: Boolean);

var

  s, a: string;

  i: integer;

begin

  s := '';

  a := '';

  for i:=low(Fields) to high(Fields) do

  begin

    s := s + a + Fields[i].FieldName + ' to ' + Fields[i].AsString;

    a := ',';

  end;

  ShowMessage('The following fields have changed their value:' + s);

end;

Fields 배열이 실제로는 데이터셋의 필드들이 아니라 에러테이블의 필드들을 가리킨다는 것을 주의하십시오. 따라서, 원래의 필드 값을 얻어오려면 다음과 같이 하면 됩니다.

Originalvalue := yourquery.FieldByName(Fields[i].FieldName).AsString;

Message 인자는 리졸브 작업중에 필드 값이 변경될 때 제공된 메시지를 가지고 있습니다. Skip 인자는 디폴트로 false이며, 현재 레코드에 대한 모든 변경 사항을 무시하려는 목적으로 true로 설정할 수 있습니다. 이것은 사용자의 disconnected 데이터셋의 수정된 내용이 반영되지 않는다는 것을 의미합니다.
리졸브 작업중에 다른 데이터베이스 작업을 하기
특별한 상황에서라면 리졸빙 작업중에 다른 테이블을 업데이트할 필요가 있을 수도 있습니다.
이렇게 하려면, 현재 데이터베이스 작업을 위해 사용되고 있는 연결을 가리키는 리졸버의 Connection 프로퍼티를 이용할 수 있습니다. 이 연결을 이용하므로, 여러분이 데이터베이스에 하는 모든 업데이트 작업은 완전한 트랜잭션의 일부가 됩니다.

procedure TForm1.kbmMWBDEResolver1Delete(Sender: TkbmMWCustomResolver;

  var Skip, Abort: Boolean);

var

  c: TkbmMWBDEConnection;

  q: TQuery;

begin

  c := TkbmMWBDEConnection(Sender.Connection);

  q := TQuery.Create;

  try

    q.Database := c.Database;

    q.SQL := 'INSERT INTO TRACE (MESSAGE) VALUES (''DELETING '

      + ValueByName['somefield'] + ''')';

    q.ExecSQL;

  finally

    q.Free;

  end;

end;
자동 증가 필드의 리졸빙
리졸버의 UpdateInsertAutoIncFields 속성의 기본 설정에서는 자동 증가(Autoincrement) 필드가 리졸브되지 않습니다.

하지만 자동증가 필드는 레코드가 인서트될 때 데이터베이스에 의해 자동으로 새 유니크한 값이 할당됩니다. KbmMW는 

However while the record is inserted, the autoincrement field will automatically be assigned a new unique value by the database. kbmMW supports getting that value for a good part of the supported databases (although not guaranteed for all).

레코드가 인서트될 때, 해당 레코드에 해당하는 새 autoinc 값을 가져오기 위해 다음과 같은 리졸버 프로퍼티를 사용할 수 있습니다. (테이블당 단 하나의 autoinc 필드만 지원됩니다).

    property LastAutoIncValueByIndex[AIndex: integer]: integer

    property LastAutoIncValueByTableName[ATableName: string]: integer

LastAutoIncValueByIndex require you to provide an index from 0 to the number of tables being resolved less one (see resolving joins).

LastAutoIncValueByTableName allow you to provide the table name for which you want to get the latest autoinc value. The table obviously must be part of the tables being resolved.

The resolver automatically logs field changes when an autoinc field change value during resolve.
When we later talk about resolving joins, we will see how to use this to effectively resolve master/detail relations which involves autoinc fields.
필드 선택적인 리졸빙
그럼, 필드들 중 일부가 리졸브되어서는 안되는 경우에는 어떻게 해야 할까요? 예를 들어,

SELECT * from customer order by custno

Contact 라는 필드가 리졸브되지 않기를 바라는 경우, 혹은 이 필드가 다른 두 필드의 값을 더한 값을 가지고 있는 경우일 수도 있습니다.

특정 필드가 리졸브되지 않도록 하기 위한 몇가지 방법이 있습니다.

1) customer 테이블에서 리졸브되어서는 안되는 필드들만 제외한 모든 필드들의 Origin 속성을 설정합니다. 리졸버의 SkipFieldsWithoutOrigin 속성을 true로 설정합니다.
2) 필드의 ProviderFlags 속성에 pfInUpdate 값을 포함시킵니다. 이렇게 하면 This way you can also control if a field should be considered part of the unique key or not during resolve by setting pfInKey and pfInWhere.

3) Use the database adapter’s resolver’s event ExcludeFromUpdateInsert and ExcludeFromWhere. Simply return a set of matching field provider flags in the Flags argument of the event excluding the operation you don’t want the field to participate in. The returned set can contain mwpfInUpdate, mwpfInInsert and mwpfInKey.

Further the resolve have several properties which can be set to include or exclude resolving fields marked as readonly in the dataset, key fields and autonumeric fields namely UpdateInsertReadonlyFields, UpdateInsertAutoIncFields, InsertKeyFields, UpdateKeyFields.
조인된 결과의 리졸빙
이제, 좀 더 고급 셀렉트 문으로 프로젝트를 진행시켜봅시다.

SELECT * from customer,orders where orders.custno=customer.custno order by customer.custno

위 SQL문은 두 테이블을 조인한 것으로, 이런 경우 리졸브 작업을 좀 더 복잡해집니다.

위의 SQL문이 실행되면 customer 테이블과 orders 테이블로부터 모든 필드들이 하나의 레코드로 리턴됩니다. 몇몇 필드 이름은 두 테이블에서 서로 충돌합니다. 예를 들어, custno와 taxrate는 둘 다 customer 테이블과 orders 테이블에 존재합니다. 결과셋에서는 같은 이름을 갖는 두 필드가 존재할 수 없으므로, 유일하지 않은 두 필드 이름들은 자동으로 유일한 이름이 주어집니다. 따라서 결과셋에는customer.custno, customer.taxrate, orders.custno_1, orders.taxrate_1 필드가 리턴되게 됩니다.

custno_1과 taxrate_1가 원래의 테이블에 존재하지 않으므로 이런 경우 직접 리졸브하려고 하면 제대로 동작하지 않습니다. 따라서, 필드가 어느 테이블의 어느 필드로부터 유래된 것인지를 알리기 위해 필드의 오리진을 지정해주어야 합니다. 

ADOX 어댑터 같은 일부 데이터베이스 어댑터들은 자동으로 이 오리진 정보를 설정해주지만, BDE같은 다른 어댑터들은 그렇지 않습니다. 데이터베이스 API가 그런 정보를 지원하지 않기 때문입니다.

이것은, 개발자가 직접 오리진을 지정해주어야 한다는 것을 의미합니다.

리졸브되기 전에 리졸브될 모든 필드들에 오리진 값이 지정되어 있어야 합니다. 물론 이 작업은 단 하나의 테이블만이 관계되어 있을 때는 필요하지 않습니다. 리턴된 필드 이름이 데이터베이스 테이블의 실제 필드 이름과 일치하기 때문입니다.

우리 예제에서는, 위의 SQL 문을 TkbmMWBDEQuery 컴포넌트의 SQL 프로퍼티에 넣습니다.

다음으로 TkbmMWBDEQuery 컴포넌트를 더블클릭하여 필드 에디터를 띄웁니다.
오른쪽 클릭하고 ‘Add all fields’를 선택합니다.
이렇게 하면 쿼리 컴포넌트의 모든 필드가 정적으로 생성됩니다. 동일한 쿼리 컴포넌트를 여러 다른 타입의 쿼리에 사용한다면, 코드로 런타임에 필드 오리진을 설정해줄 필요가 있습니다.

필드들을 정적으로 생성했다면 필드 오리진을 디자인타임에 설정할 수 있습니다. Shift나 Ctrl 키를 누르고 필드를 클릭하면 필드 디자이너에서 동시에 여러 필드들을 선택할 수 있습니다.

Customer 테이블에서 나온 모든 필드들을 선택하십시오. 그리고 F11 키를 눌러 오브젝트 인스펙터를 띄웁니다.

Origin 속성에 customer. 라고 써넣습니다.
끝에 따라붙는 마침표(.)에 주의하십시오!

이것은 이 필드들이 ‘customer’ 테이블에서 나온 것이라는 것을 의미합니다. 끝에 마침표가 없으면 kbmMW는 해당 필드가 ‘customer’라는 데이터베이스 필드로부터 나왔다고 인식하게 됩니다.

Origin 값이 마침표로 끝나면, 데이터베이스 테이블의 필드 이름이 TField 객체의 필드 이름과 같다는 것을 의미합니다.

다음으로, Orders 테이블에서 나온 모든 필드들을 선택하고 그 Origin 값을 orders. 라고 설정합니다.

이번에도 끝에 마침표를 붙이는 것을 잊지 마십시오.

이제 우리는 쿼리가 자동으로 유일 이름을 붙여준 필드들에 대해 별도의 처리를 해주어야 합니다.

필드 리스트에서 CustNo_1라는 이름의 필드를 선택합니다. Origin 속성의 값을 orders.custno로 설정합니다.

필드 리스트에서 TaxRate_1라는 이름의 필드를 선택합니다. Origin 속성의 값을 orders.taxrate 로 설정합니다.

이제 우리는 필드 객체가 어느 데이터베이스 필드로부터 나왔는지 지정해주었습니다.

다음 단계는 ‘Simple resolving’에서 했던 것처럼 TableName과 KeyFieldNames에 대한 정보를 지정해주는 것입니다.

이 예제에서 우리는 customer의 필드들과 orders 필드들을 모두 리졸브하려고 합니다.

따라서 다음과 같이 설정합니다.

TableName := 'customer;orders'

이렇게 하면 리졸버는 두 테이블이 리졸브되어야 한다는 것을 알 수 있게 됩니다. TableName 속성에 나열된 테이블들에 속한 필드들만이 리졸브됩니다. 따라서 조인 자체는 완벽하게 유효하지만, 조인에 관계된 테이블들 중 하나나 일부만 리졸브할 수 있게 됩니다.

마지막으로 다음과 같이 설정합니다.

KeyFieldNames := 'customer.custno;orders.orderno'

이것은 테이블에서 유일한 키 필드로 알려진 필드를 지정하는 것입니다.

한 테이블에서 유일 키가 두 필드로 이루어져 있다면, 두 필드 이름을 테이블 이름을 포함하여 지정하면 됩니다.

리졸버의 SkipFieldsWithoutOrigin 속성은 오리진이 없는 필드가 어디로 갈지 혼란이 생길 수 있기 때문에 보통은 true로 설정됩니다. 모든 필드들의 Origin 값이 설정되어 있다면, SkipFieldsWithoutOrigin 속성은 아무런 영향이 없게 됩니다.

조인의 리졸브 작업은 완전한 트랜잭션의 제어 아래에서 수행됩니다. 이것은, 한 테이블에 대한 리졸브에서 에러가 발생한다면 모든 테이블에 대한 모든 업에이트가 롤백된다는 것을 의미합니다.

여기까지가 조인 결과를 리졸브하기 위해 알아야 할 모든 것들입니다.

마스터/디테일 관계의 핵심은 

The keys for the master/detail relation are manually determined by the user or application when adding new records.

하지만 키 필드가 자동증가 타입이라면 어떨까요? 그런 경우라면, customer 테이블에 대한 disconnected 사용자 데이터셋에 제공된 값들은 데이터베이스에 저장되는 값들과 다를 것입니다. 이것은 새 customer 레코드를 인서트할 때 데이터베이스가 자동증가 (키) 필드에 자동으로 새 유일 숫자를 할당하기 때문입니다.

그런 경우에, customer 테이블(마스터 테이블)의 해당 레코드를 가리키는 orders 테이블(디테일 테이블)의 필드가 매치되지 않게 될 것입니다.

그에 대한 해결책도 있습니다.

Make sure that the master table is listed first in the TableName property, followed by the detail table. E.g. q.TableName := 'customer;orders'.

Doing that ensures that fields belonging to the customer (master table) is resolved first, followed by fields from the orders (detail) table.

Because of that we can use the LastAutoIncValue…methods to obtain the correct newly created autoinc value for the customer table and use that value as the reference value in the orders table. E.g.

procedure TForm1.kbmMWBDEResolver1GetValue(

  ADeltaHandler: TkbmCustomDeltaHandler; AField: TField;

  var AValue: Variant);

begin

  with TkbmMWCustomResolver(ADeltaHandler) do
  begin

    if (State=mwrsInsert) and (AField.FieldName='CustNo') then
    begin

      Avalue := LastAutoIncValueByTableName['customer'];

      LogFieldValueChange([AField], [AValue],

'Referenced customer record was changed to ' + AValue);

    end;

  end;

end;

At the same time we remember to log the change so the disconnected dataset can be automatically updated with the new valid values without having to reopen the query.
여러 서버 데이터셋을 하나의 트랜잭션으로 리졸브하기
여러분이 여러 개의 데이터셋을 가지고 있고, 

If you have multiple datasets, possibly originating from different database servers of even different types, and you resolve the changes in each dataset, one by one, you can have the problem that one dataset is resolved ok, and another is not which may leave you with a set of partially updated databases.

To remedy that, kbmMW employs a server side (db adapter side) component called TkbmMWTransactionResolver.

This component handle distributed transaction control. It means that it automatically will start, commit or rollback transactions on multiple databases at the same time, virtually operating like the updates on all the databases are all running in one big transaction.

It’s very easy to use. Add an instance of TkbmMWTransactionResolver 

 to your application at design time, or create one at runtime. Eg.

var

  t: TkbmMWTransactionResolver;

begin

   t := TkbmMWTransactionResolver.Create(nil);

   try

     t.Resolve([dataset1, dataset2, …, datasetn]);

   finally

     t.Free;

   end;

end; 

Each dataset will automatically be resolved according to the normal resolving settings used when resolving the dataset standalone. The difference with the transaction resolver is that if one of the datasets does not resolve properly, all the resolves for all datasets will be rolled back. It however requires that the backend database fully supports transactions.

What if you want not only to resolve changes in datasets, but also do other custom SQL operations within the same transaction?

That’s simple.

We add a query component containing the SQL you want to introduce to your database while resolving. E.g.

qSpecialStuff.SQL.Text := 'INSERT INTO sometable (X,Y,Z) VALUES (10,22,33)';

이제, TransactionOperation 속성을 mwtoExecute로 설정합니다(기본값은 mwtoResolve입니다). 그리고 다음과 같이 qSpecialStuff을 트랜잭션 리졸버에서 리졸브할 데이터셋으로 포함시킵니다.

t.Resolve([dataset1, dataset2, qSpecialStuff, …, datasetn]);

이제 트랜잭션 리졸버는 dataset1과 dataset2을 이 순서대로 리졸브하고, 다음으로 qSpecialStuff에서 지정한 SQL문을 실행시키고 그 다음의 데이터셋들을 리졸브를 계속합니다.

여러분은 트랜잭션 리졸브 작업 중 어디에서든 원하는 만큼의 SQL문을 사용할 수 있습니다.

N-티어 클라이언트 사이드 리졸빙
Keeping the knowledge from the previous chapters present, we will now resolve data in an n-tier setup from the client. The interesting part is that now we have an application server in-between through which the client communicates.

애플리케이션 서버

First create an application server containing a query service following the guidelines from ‘Using kbmMW as a query server’ whitepaper, or use one of the predefined ones like the BDE application server.

The query service could look like this (as copied from the BDE server):

The query components all have the Resolver property hooked up to the kbmMWBDEResolver1 component.

This query service contains 3 query components where two of them contain a predefined SQL statement while one (ClientSideQuery) require the client to provide the SQL statements.

In case of SELECTED_EVENT and ALL_EVENTS, the TableName and KeyFieldNames properties should be given at server side. This makes sure that the client do not have to worry about that when the client want to resolve.

Also field Origins and/or Providerflags should be specified on the server side for those two queries if they are needed at all (eg. in join situations etc.). The client will automatically get to know the values defined on the server.

The ClientSideQuery is another story. Since the client can give any SQL statement it cares for, predefining TableName, KeyFieldNames, Origins and ProviderFlags wont make sense on the server. Its the responsibility of the client to setup that.

To let the client provide TableName and KeyFieldNames settings, a couple of properties must be set on the query service authorizing the client to do so.

AllowClientKeyFields and AllowClientTableName must be true. In addition, AllowClientStatement should be true to let the client specify a query statement (SQL) of its own.

AllOrNothing on the resolver component should usually also be set to true to disallow partial successful resolves.

The client application

The client contains a TkbmMWClientTCPIPIndyTransport, a TkbmMWClientConnectionPool, a TkbmMWPooledSession, a TkbmMWClientQuery and a dataset streamformatter in addition to the dataaware controls.

All components are hooked together according to the ‘Using kbmMW as a query server’ whitepaper.

If the client should access the predefineded server query ALL_EVENTS, the query statement in the client should be ‘@ALL_EVENTS’ utilizing a named query statement.

When opening the query, all informations needed to resolve are automatically transferred from the server to the client so that the client can advice the server about them later on when/if the client choose to resolve changes in the resultset.

In other words, all what the developer needs to do is to open the query, let the user alter data in the grid, and press a key which calls the Resolve method of the client query component.

If the client will itself provide query/SQL statements, its now the responsibility of the client to setup TableName, KeyFieldNames, optionally field Origins and ProviderFlags.

In some cases, the server is able to deliver correct Origin values, but this is very much dependent on the database and its API as mentioned earlier on.

Error handling after a resolve event on the client is exactly the same as on the server, described in a previous chapter.

여러 클라이언트 데이터셋을 하나의 트랜잭션으로 리졸브하기
앞에서, 서버 사이드에서 여러 데이터셋을 완전한 트랜잭션 컨트롤 하에서 하나의 단일 작업으로 리졸브 작업을 하기 위한 TkbmMWTransactionResolver 컴포넌트를 살펴봤습니다. 클라이언트에서 같은 기능을 하는 컴포넌트로 TkbmMWClientTransactionResolver 

 컴포넌트가 있습니다.

이 컴포넌트를 이용하려면 다음과 같은 속성들이 설정되어야 합니다.

ConnectionPool
TkbmMWClientConnectionPool 컴포넌트를 가리켜야 합니다.
QueryService

App 서버에서 리졸빙을 책임져야 할 쿼리 서비스 이름으로 설정합니다.
QueryServiceVersion
쿼리 서비스에 해당하는 버전 문자열로 설정합니다.
TransportStreamFormat
서버 사이드의 트랜스포트와 맞는 데이터셋 트랜스포트 포맷 컴포넌트로 설정합니다. (예를 들면 TkbmMWBinaryStreamFormat)
이제 필요한 것은 클라이언트 데이터셋들을 리졸브하기 위해 Resolve 메소드를 호출하는 것 뿐입니다.

Myclienttransactionresolver.Resolve([clientdataset1,…, clientdatasetN]);

만약 리졸브 호출이 false를 리턴하면 리졸브 작업이 실패한 것이며, 그 이유를 알아내기 위해서는 ErrorTable을 살펴봐야 합니다. (혹은 OnResolveError 이벤트를 이용할 수도 있습니다)
서버 사이드 트랜잭셔널 리졸빙 컴포넌트와 마찬가지로, 별도의 SQL문을 실행하기 위해서는 데이터셋 리스트에 해당 SQL문을 포함하는 데이터셋을 포함시키고 TransactionOperation을 mwtoResolve 대신 mwtoExecute으로 설정하면 됩니다.
커스텀 리졸버 개발하기
아주 특별한 몇몇 경우에는 리졸버의 작업에 대해 전반적으로 제어할 필요가 있을 수도 있습니다. 예를 들면, 데이터베이스가 아니라 파일 시스템 등 다른 데이터 구조에 리졸빙 작업을 해야 할 경우가 그렇습니다.

이런 경우, 직접 리졸버 컴포넌트를 만드는 것이 쉬울 수 있습니다. 다음은 간단한 리졸버의 예입니다.
TYourResolver = class(TkbmMWCustomResolver)

protected

  function DoInsertRecord(AObject: TkbmMWCustomResolverObject): boolean; override;

  function DoModifyRecord(AObject: TkbmMWCustomResolverObject): boolean; override;

  function DoDeleteRecord(AObject: TkbmMWCustomResolverObject): boolean; override;

end;

function TYourResolver.DoInsertRecord(AObject: TkbmMWCustomResolverObject)
: boolean;

var

  i: integer;

  v: variant;

  s: string;

begin

  // Getting new values for the record to be inserted.

  for i:=0 to FieldCount-1 do
  begin

    v := Values[i];     // The value for a field.

    S := FieldNames[i]; // s contain the name for the field.

    // Do whatever you need here.

    ...

    // If you raise an exception it will automatically add the exception message to 

    // the ErrorTable and continue with next record, unless AllOrNothing is true.

  end;

  // If you don’t return true, an error record will be added to the ErrorTable.

  Result := true;

end;

function TYourResolver.DoModifyRecord(AObject: TkbmMWCustomResolverObject)
: boolean;

begin

     Result := true;

end;

function TYourResolver.DoDeleteRecord(AObject: TkbmMWCustomResolverObject)
: boolean;

begin

  Result := true;

end;

Then simply assign an instance of the resolver to the Resolver property of the server query/stored procedure component. You can register the resolver as a component to be able to use it at designtime just like any other custom component.

이것으로 리졸브 문서를 마치겠습니다.
Kim Madsen

Components4Developers

